

# Beágyazott nyelvek

# Bevezetés

Domain-specific language (DSL), „szakterület-specifikus nyelv”:

*„A computer programming language of limited expressiveness focused on a particular domain.”* – Martin Fowler, *Domain-Specific Languages*, 2011

Például: 4GL nyelvek, SQL, UNIX-os segédprogramok: shellek, reguláris kifejezések, *lex(1)*, *yacc(1)*.

- ▶ Az adott (szak)területhez tartozó fogalmak könnyebben megfoghatóak, közvetlenebb és egységesebb formában fejezhetőek ki.
- ▶ Ellenben minden más a (szak)területen kívül nehezen írható le.
- ▶ Nem feltétlenül Turing-teljes.
- ▶ Haskell = DSL for compilers :-)

# DSL-ek megvalósítása

DSL-eket kétféleképpen szoktak megvalósítani:

- ▶ *Önálló (standalone)*: Saját szintaxissal rendelkezik, a végrehajtáshoz a hagyományos fordítási technikák segítségével fordítjuk a DSL programokat általános célú programokra.
  - ▶ A szintaxis a célközönség számára alakítható.
  - ▶ Külön elemző, fordító és fejlesztői környezet kell.
  - ▶ Gyakran redundáns és fáradságos fejlesztés.
- ▶ *Beágyazott (embedded)*: Egy (általános célú) befogadó nyelven (host language) keresztül adjuk meg programkönyvtárként, makrócsomagként.
  - ▶ embedded domain-specific language, EDSL

# DSL-ek megvalósítása

DSL-eket kétféleképpen szoktak megvalósítani:

- ▶ *Önálló (standalone)*: Saját szintaxissal rendelkeznek, a végrehajtáshoz a hagyományos fordítási technikák segítségével fordítjuk a DSL programokat általános célú programokra.
- ▶ *Beágyazott (embedded)*: Egy (általános célú) befogadó nyelven (host language) keresztül adjuk meg programkönyvtárként, makrócsomagként.
  - ▶ Az így rendelkezésre álló eszközök felhasználhatóak, beleértve a szintaktikai konvenciókat is.
  - ▶ Ha a befogadó nyelv merev, akkor a szintaxis kényelmetlen tud lenni.
  - ▶ Nehéz meghúzni az EDSL és befogadó nyelv közti határt.
  - ▶ A hibaüzenetek olykor semmitmondóak.

# DSL-ek beágyazása

A funkcionális nyelvek elég jól használhatóak beágyazásra:

- ▶ Magasabb rendű függvények, sémák
- ▶ Lusta kiértékelés, végtelen adatszerkezetek
- ▶ Gazdag (programozható!) típusrendszer, típuskikövetkeztetés

Nyelvbeágyazási stílusok:

- ▶ *Mély beágyazás (deep embedding)*: A DSL egyes szerkezeteit megvalósító függvények egy absztrakt szintaxisfát (AST) építenek fel, amelyet transzformálhatunk, futtatunk (algebra).
- ▶ *Sekély beágyazás (shallow embedding)*: A DSL egyes szerkezeteit közvetlenül a neki megfelelő függvényekkel fejezzük ki, nincs AST, sem bejárás.
- ▶ Az előbbi kettő valamilyen kombinációja.

# Példa: BASIC Haskellben

-- <http://hackage.haskell.org/package/BASIC>

**import** *Language.BASIC*

*main* :: IO ()

*main* = runBASIC \$ do

10 GOSUB 1000

20 PRINT " \* Welcome to HiLo \* "

30 GOSUB 1000

100 LET I := INT(100 \* RND(X))

200 PRINT " Guess my number : "

210 INPUT X

220 LET S := SGN(I - X)

230 IF S <> 0 THEN 300

240 FOR X := 1 TO 5

250 PRINT X \* X; " You won!"

260 NEXT X

270 STOP

300 IF S <> 1 THEN 400

310 PRINT " Your guess "; X; " is too low."

320 GOTO 200

400 PRINT " Your guess "; X; " is too high."

410 GOTO 200

1000 PRINT " \* \* \* \* \* "

1010 RETURN

9999 END

## További példák EDSL-ekre

```
-- http://hackage.haskell.org/package/haskelldb  
import Database.HaskellDB
```

```
query = do  
  cust ← table customers  
  restrict (cust ! city . == . "Budapest")  
  project (cust ! customerID)
```

```
-- http://hackage.haskell.org/package/uu-parsinglib  
import Text.ParserCombinators.UU
```

```
expr = Let <$ keyword "let" <*> decl <*>  
        keyword "in" <*> expr  
        <|> operatorExpr
```

```
-- https://hackage.haskell.org/package/accelerate  
import Data.Array.Accelerate
```

```
dotp :: Acc (Vector Float) → Acc (Vector Float) → Acc (Scalar Float)  
dotp xs ys = fold (+) 0 (zipWith (*) xs ys)
```

# Egy nagyon egyszerű DSL

Készítsünk egy DSL-t, amely egész számok véges halmazait írja le!

**type** *IntegerSet*

*empty* :: *IntegerSet*

*insert* :: *Integer* → *IntegerSet* → *IntegerSet*

*delete* :: *Integer* → *IntegerSet* → *IntegerSet*

*member* :: *Integer* → *IntegerSet* → *Bool*

Példa az alkalmazásra:

$x = \text{member } 3 (\text{insert } 1 (\text{delete } 3 (\text{insert } 2 (\text{insert } 3 \text{ empty}))))$



# Egy nagyon egyszerű DSL – mint könyvtár

Az egész számok halmaza tulajdonképpen könyvtárként is megvalósítható, a kapcsolódó típusok, függvények gyűjteményeként.

```
type IntegerSet = [Integer]
```

```
empty :: IntegerSet  
empty = []
```

```
insert :: Integer → IntegerSet → IntegerSet  
insert = (:)
```

```
delete :: Integer → IntegerSet → IntegerSet  
delete x = filter (≠ x)
```

```
member :: Integer → IntegerSet → Bool  
member x = any (≡ x)
```

```
-- Alkalmazás:
```

```
x = 3 `member` (1 `insert` (3 `delete` (2 `insert` (3 `insert` empty))))
```

A reprezentáció nem rejtett.

# Egy nagyon egyszerű DSL – mint modul

```
module SetModule (Set, runSet, insert, delete, member) where  
import Control.Monad.State
```

```
type IntegerSet = [Integer]
```

```
newtype Set  $\alpha$  = S (State IntegerSet  $\alpha$ )  
    deriving (Functor, Applicative, Monad)
```

```
runSet :: Set  $\alpha$   $\rightarrow$   $\alpha$   
runSet (S s) = evalState s []
```

```
insert :: Integer  $\rightarrow$  Set ()  
insert x = S $ modify (x :)
```

```
delete :: Integer  $\rightarrow$  Set ()  
delete x = S $ modify (filter ( $\neq$  x))
```

```
member :: Integer  $\rightarrow$  Set Bool  
member x = S $ get >>= return  $\circ$  any ( $\equiv$  x)
```

-- Alkalmazás:

```
x = runSet $ do { insert 3; insert 2; delete 3; insert 1; member 3 }
```

# Egy nagyon egyszerű DSL – mint absztrakt adattípus

```
module SetADT (IntegerSet, empty, insert, delete, member) where
```

```
newtype IntegerSet = IS { unIS :: [Integer] }
```

```
empty :: IntegerSet
```

```
empty = IS []
```

```
insert :: Integer → IntegerSet → IntegerSet
```

```
insert x = IS ∘ (x :) ∘ unIS
```

```
delete :: Integer → IntegerSet → IntegerSet
```

```
delete x = IS ∘ filter (≠ x) ∘ unIS
```

```
member :: Integer → IntegerSet → Bool
```

```
member x = any (≡ x) ∘ unIS
```

```
-- Alkalmazás:
```

```
x = 3 `member` (1 `insert` (3 `delete` (2 `insert` (3 `insert` empty))))
```

Teljesen funkcionális stílus: nincs monád, mindig új halmazokat szerkesztünk.

# Egy nagyon egyszerű DSL – mélyen beágyazva

```
module SetDeep (IntegerSet (Empty, Insert, Delete), member) where
```

```
data IntegerSet
```

```
  = Empty
```

```
  | Insert Integer IntegerSet
```

```
  | Delete Integer IntegerSet
```

```
member :: Integer → IntegerSet → Bool
```

```
_ `member` Empty      = False
```

```
x `member` (y `Insert` ys) = (x ≡ y) ∨ x `member` ys
```

```
x `member` (y `Delete` ys) = (x ≠ y) ∧ x `member` ys
```

```
-- Alkalmazás:
```

```
x = 3 `member` (1 `Insert` (3 `Delete` (2 `Insert` (3 `Insert` Empty))))
```

# Egy nagyon egyszerű DSL – sekélyen beágyazva

**module** SetShallow (*IntegerSet*, *empty*, *insert*, *delete*, *member*) **where**

**newtype** *IntegerSet* = *IS* (*Integer*  $\rightarrow$  *Bool*)

*empty* :: *IntegerSet*

*empty* = *IS* \$  $\lambda y$  . *False*

*insert* :: *Integer*  $\rightarrow$  *IntegerSet*  $\rightarrow$  *IntegerSet*

$x$  `insert` (*IS* *f*) = *IS* \$  $\lambda y$  . ( $x \equiv y$ )  $\vee$  *f* *y*

*delete* :: *Integer*  $\rightarrow$  *IntegerSet*  $\rightarrow$  *IntegerSet*

$x$  `delete` (*IS* *f*) = *IS* \$  $\lambda y$  . ( $x \not\equiv y$ )  $\wedge$  *f* *y*

*member* :: *Integer*  $\rightarrow$  *IntegerSet*  $\rightarrow$  *Bool*

$x$  `member` (*IS* *f*) = *f* *x*

-- Alkalmazás:

$x$  = 3 `member` (1 `insert` (3 `delete` (2 `insert` (3 `insert` *empty*))))

# Egy nagyon egyszerű DSL – összefüggés

A sekély beágyazást a mély beágyazásból úgy kapjuk, ha felszámoljuk a köztes adatszerkezetet, vagyis az AST-t.

```
elements :: SetDeep.IntegerSet → SetADT.IntegerSet  
elements Empty = SetADT.empty  
elements (x `Insert` xs) = x `SetADT.insert` (elements xs)  
elements (x `Delete` xs) = x `SetADT.delete` (elements xs)  
membership :: SetADT.IntegerSet → SetShallow.IntegerSet  
membership xs = SetShallow.IS $ λ x . any (≡ x) (SetADT.unIS xs)
```

# Az implementációs technikák összehasonlítása

Mély beágyazás: a *legabsztraktabb* implementáció.

- ▶ Könnyű új értelmezőket (*observer*), nehéz viszont új műveleteket készíteni.
- ▶ Optimalizálás és konvertálás feldolgozás közben.
- ▶ Az AST vizualizálható és viselkedése nyomkövethető.
- ▶ Nehezebb kihasználni a közös részkifejezések megosztását (*sharing*, ld. később bővebben).

Sekély beágyazás: a *legkonkrétabb* implementáció.

- ▶ Nehéz új értelmezőket, könnyű viszont új műveleteket készíteni.
- ▶ Közvetlenül denotációs szemantikát adunk meg.
- ▶ Viszonylag könnyen elérhetőek a Haskell lehetőségei (pl. rekurzió, *sharing*).
- ▶ Nehezen látható a viselkedése az egyetlen értelmező miatt.

# Funkcionális előnyök: Típusok

- ▶ Az algebrai adattípusok kényelmesen használhatóak mélyen beágyazott nyelvek absztrakt szintaxisfáinak modellezésére:
  - ▶ Az értelmezők, esetenként a műveletek rekurzív függvényekként írhatóak meg, a szerkezet feletti indukcióval.
  - ▶ Az optimalizációk és transzformációk átrendezésekkel valósíthatóak meg.
  - ▶ Nagyon hasznos köztes forma integráció esetén, amikor szövegből hozzuk létre, és szövegre alakítjuk át.
- ▶ Különbféle számítások írhatóak le ezen a módon ( $IO\ \alpha$ ), amelyek valamilyen konkrét Haskell-típushoz ( $\alpha$ ) kötődnek.
- ▶ Teljesen irányítható, hogy miként hozzuk létre, kombináljuk és számítjuk ki ezeket a számításokat.



# Funkcionális előnyök: Típusok (példa)

Tekintsük a következő interfészt, nyelvet:

*mkInt*      $:: \text{Int} \rightarrow X \text{ Int}$

*mkChar*    $:: \text{Char} \rightarrow X \text{ Char}$

*combine*    $:: X \alpha \rightarrow X \alpha \rightarrow X \alpha$

*pair*        $:: X \alpha \rightarrow X \beta \rightarrow X (\alpha, \beta)$

*run*        $:: X \alpha \rightarrow \alpha$

- ▶ *X Bool* sosem hozható létre.
- ▶ Ennek az EDSL-nek a típuskészlete kisebb.
- ▶ Az így beágyazott értékekhez a nekik megfelelő, optimális ábrázolást vagy számítási módot társíthatunk.

# Funkcionális előnyök: Típusok (példa)

```
import Control.Monad.ST  
import Data.STRef
```

$newSTRef :: \alpha \rightarrow ST\ \sigma\ (STRef\ \sigma\ \alpha)$

$readSTRef :: STRef\ \sigma\ \alpha \rightarrow ST\ \sigma\ \alpha$

$writeSTRef :: STRef\ \sigma\ \alpha \rightarrow \alpha \rightarrow ST\ \sigma\ ()$

$runST :: (\forall\ \sigma . ST\ \sigma\ \alpha) \rightarrow \alpha$

- ▶ A feladat:
  - ▶ Az erőforrás lefoglalása egy számításban
  - ▶ A legfoglalt erőforrás átadása egy másik számításnak
  - ▶ Az átadott erőforrás használata ott
- ▶ A megoldás (típusrendszerrel):
  - ▶ A számításokat és az erőforrásokat régiókkal paraméterezzük (mint egzisztenciális típus)
  - ▶ A számítások nem tudhatják, melyik régióban futnak
  - ▶ Az átadott hivatkozások által függés alakul ki

# Funkcionális előnyök: Típusok (példa)

```
data Expr
  = Val Integer
  | ValB Bool
  | Add Expr Expr
  | And Expr Expr
  | Eq Expr Expr
  | If Expr Expr Expr

eval :: Expr → Either Integer Bool
eval (Val n) = Left n
eval (ValB b) = Right b
eval (x `Add` y) =
  case (eval x, eval y) of
    (Left m, Left n) → Left (m + n)
eval (x `And` y) =
  case (eval x, eval y) of
    (Right a, Right b) → Right (a ∧ b)
eval (x `Eq` y) =
  case (eval x, eval y) of
    (Left m, Left n) → Right (m ≡ n)
    (Right m, Right n) → Right (m ≡ n)
eval (If x y z) =
  case (eval x) of
    Right b → if b then (eval y) else (eval z)
```

# Funkcionális előnyök: Típusok (példa)

```
{-# LANGUAGE GADTs, KindSignatures #-}
```

```
data Expr ::  $\star$  where
```

```
  Vall :: Integer  $\rightarrow$  Expr
```

```
  ValB :: Bool  $\rightarrow$  Expr
```

```
  Add  :: Expr  $\rightarrow$  Expr  $\rightarrow$  Expr
```

```
  And  :: Expr  $\rightarrow$  Expr  $\rightarrow$  Expr
```

```
  Eq   :: Expr  $\rightarrow$  Expr  $\rightarrow$  Expr
```

```
  If   :: Expr  $\rightarrow$  Expr  $\rightarrow$  Expr  $\rightarrow$  Expr
```

```
eval :: Expr  $\rightarrow$  Either Integer Bool
```

```
eval (Vall n) = Left n
```

```
eval (ValB b) = Right b
```

```
eval (x `Add` y) =
```

```
  case (eval x, eval y) of
```

```
    (Left m, Left n)  $\rightarrow$  Left (m + n)
```

```
eval (x `And` y) =
```

```
  case (eval x, eval y) of
```

```
    (Right a, Right b)  $\rightarrow$  Right (a  $\wedge$  b)
```

```
eval (x `Eq` y) =
```

```
  case (eval x, eval y) of
```

```
    (Left m, Left n)  $\rightarrow$  Right (m  $\equiv$  n)
```

```
    (Right m, Right n)  $\rightarrow$  Right (m  $\equiv$  n)
```

```
eval (If x y z) =
```

```
  case (eval x) of
```

```
    Right b  $\rightarrow$  if b then (eval y) else (eval z)
```

# Funkcionális előnyök: Típusok (példa)

{-# LANGUAGE GADTs, KindSignatures #-}

**data** *Expr* ::  $\star \rightarrow \star$  **where**

*Vall* :: *Integer*  $\rightarrow$  *Expr Integer*

*ValB* :: *Bool*  $\rightarrow$  *Expr Bool*

*Add* :: *Expr Integer*  $\rightarrow$  *Expr Integer*  $\rightarrow$  *Expr Integer*

*And* :: *Expr Bool*  $\rightarrow$  *Expr Bool*  $\rightarrow$  *Expr Bool*

*Eq* :: *Eq*  $\alpha \Rightarrow$  *Expr*  $\alpha \rightarrow$  *Expr*  $\alpha \rightarrow$  *Expr Bool*

*If* :: *Expr Bool*  $\rightarrow$  *Expr*  $\alpha \rightarrow$  *Expr*  $\alpha \rightarrow$  *Expr*  $\alpha$

*eval* :: *Eq*  $\alpha \Rightarrow$  *Expr*  $\alpha \rightarrow$   $\alpha$

*eval* (*Vall* *n*) = *n*

*eval* (*ValB* *b*) = *b*

*eval* (*x* `Add` *y*) = (*eval* *x*) + (*eval* *y*)

*eval* (*x* `And` *y*) = (*eval* *x*)  $\wedge$  (*eval* *y*)

*eval* (*x* `Eq` *y*) = (*eval* *x*)  $\equiv$  (*eval* *y*)

*eval* (*If* *x* *y* *z*) = **if** (*eval* *x*) **then** (*eval* *y*) **else** (*eval* *z*)

# Probléma: Rész kifejezések megosztása

Mély beágyazással elveszítjük a beágyazó nyelv bizonyos lehetőségeit, például a közös rész kifejezések megosztását.

```
toString :: Expr → String
toString (Vall n)      = show n
toString (e1 `Add` e2) =
  "(" ++ toString e1 ++ " + " ++ toString e2 ++ ")"
...
```

```
tree :: Integer → Expr
tree 0 = Vall 1
tree n =
  let shared = tree (n - 1)
  in (shared `Add` shared)
```

```
GHCi> toString (tree 3)
"(((1 + 1) + (1 + 1)) + ((1 + 1) + (1 + 1)))"
```

Hoppá! Megoldás: *Higher-Order Abstract Syntax (HOAS)*

# HOAS: A típus bővítése (egyszerűsített)

```
data Expr ::  $\star \rightarrow \star$  where  
  Val  :: Integer  $\rightarrow$  Expr  $\alpha$   
  Add  :: Expr  $\alpha \rightarrow$  Expr  $\alpha \rightarrow$  Expr  $\alpha$   
  Var  ::  $\alpha \rightarrow$  Expr  $\alpha$   
  Let  :: Expr  $\alpha \rightarrow$  ( $\alpha \rightarrow$  Expr  $\alpha$ )  $\rightarrow$  Expr  $\alpha$ 
```

A változókkal kapcsolatban semmilyen feltételezésünk nem lehet:

```
tree :: Integer  $\rightarrow$  Expr  $\alpha$   
tree 0 = Val 1  
tree n =  
  Let (tree (n - 1))  
    ( $\lambda$  shared . (Var shared) `Add` (Var shared))
```

# HOAS: Az értelmező bővítése (egyszerűsített)

```
eval :: Expr Integer → Integer
eval (Val n)      = n
eval (x `Add` y)  = (eval x) + (eval y)
eval (Var x)      = x
eval (Let e f)    = eval (f (eval e))
```

A közös részkifejezések most már valóban láthatóak (*observable*), értelmezésük mikéntje is a mienk:

```
toString :: Expr String → Int → String
toString (Val x) _ = show x
toString (x `Add` y) c = "(" ++ toString x c ++ " + " ++ toString y c ++ ")"
toString (Var x) _ = x
toString (Let e f) c =
  let v = "x" ++ show c
  in "let " ++ v ++ " = " ++ toString e (c + 1) ++
    " in " ++ toString (f v) (c + 1)
```

```
GHCi> eval (tree 2)
4
GHCi> toString (tree 2) 0
"let x0 = let x1 = 1 in (x1 + x1) in (x0 + x0)"
```



# Közös részkifejezések: Observable sharing

De használható maga a Haskell *let*?

- ▶ Lehet olyan függvényt készíteni, amellyel a részkifejezések Haskellben megjelenő megosztását lehet felismerni.  
(*"observable sharing"*)
- ▶ Ez viszont mellékhatásos, az eredmény típusa az *IO* típusba csomagolható.
- ▶ Az így megfigyelt közös részkifejezésekkel aztán tudunk építeni egy olyan adattípust, ahol szintén explicit *Let* szerepel.
- ▶ Bővebben:

<http://hackage.haskell.org/package/data-reify>

# Vissza a típusokhoz: (Függő?) Típusozott *printf()*

```
{-# LANGUAGE GADTs, KindSignatures, TypeOperators #-}
```

```
data ( $\rightarrow$ ) ::  $\star \rightarrow \star \rightarrow \star$  where
```

```
  Lit  :: String  $\rightarrow$  ( $\alpha \rightarrow \alpha$ )
```

```
  Int   ::  $\alpha \rightarrow$  (Int  $\rightarrow \alpha$ )
```

```
  Chr   ::  $\alpha \rightarrow$  (Char  $\rightarrow \alpha$ )
```

```
  ( $\vdash$ ) :: ( $\beta \rightarrow \gamma$ )  $\rightarrow$  ( $\alpha \rightarrow \beta$ )  $\rightarrow$  ( $\alpha \rightarrow \gamma$ )
```

```
infixl 5  $\vdash$ 
```

```
intp :: ( $\alpha \rightarrow \beta$ )  $\rightarrow$  (String  $\rightarrow \alpha$ )  $\rightarrow \beta$ 
```

```
intp (Lit str) c = c str
```

```
intp Int      c =  $\lambda x . c$  (show x)
```

```
intp Chr      c =  $\lambda x . c$  [x]
```

```
intp (a  $\vdash$  b) c = intp a ( $\lambda sa . intp$  b ( $\lambda sb . c$  (sa  $\vdash$  sb)))
```

```
format :: (String  $\rightarrow \alpha$ )  $\rightarrow \alpha$ 
```

```
format fmts = intp fmts id
```

```
tp1 = format (Lit "Hello world")
```

```
tp2 = format (Lit "Hello"  $\vdash$  Lit "world"  $\vdash$  Chr) '!
```

```
tp3 = format (Lit "The value of"  $\vdash$  Chr  $\vdash$  Lit "is"  $\vdash$  Int) 'x' 3
```

```
tp4 = format (Lit "abc"  $\vdash$  Int  $\vdash$  Lit "cde") 5
```

*Folytatása következik...*

# Funkcionális előnyök: Magasabb rendű függvények

A sekély beágyazás alapja a magasabb rendű függvények alkalmazása.

```
type Identifier = String
type Env       = [(Identifier, Integer)]
type Expr      = Env → Integer

val  :: Integer → Expr
var  :: Identifier → Expr
bnd  :: (Identifier, Expr) → Expr → Expr
add  :: Expr → Expr → Expr
eval :: Expr → Env → Integer

x0 = (bnd ("x", val 3) (var "x")) []
x1 = (bnd ("x", val 3) (add (var "x") (var "x"))) []
```

# Előnyök: Magasabb rendű függvények (példa)

```
type Identifier = String  
type Env       = [(Identifier, Integer)]  
type Expr      = Env → Integer
```

```
val :: Integer → Expr  
val = const
```

```
var :: Identifier → Expr  
var n = fromJust ∘ lookup n
```

```
bnd :: (Identifier, Expr) → Expr → Expr  
bnd (n, x1) x2 = λ e . x2 ((n, x1 e) : e)
```

```
add :: Expr → Expr → Expr  
add x1 x2 = λ e . (x1 e) + (x2 e)
```

```
eval :: Expr → Env → Integer  
eval = id
```

```
x0 = (bnd ("x", val 3) (var "x")) []  
x1 = (bnd ("x", val 3) (add (var "x") (var "x"))) []
```

# Funkcionális előnyök: Lusta kiértékelés

## Lusta kiértékelés:

- ▶ A függvények paramétereit csak akkor értékeljük, amikor ténylegesen szükség van rájuk („igény szerint”) – vagy egyáltalán nem, pl. a sekély beágyazás különböző értelmezései.
- ▶ Mikor kiértékelünk valamit, megtartjuk az értékét és újra felhasználjuk.
- ▶ Véges és végtelen adatszerkezetek alkalmazása is lehetséges az értelmezés során.
- ▶ Egyszerűbb programokat tudunk így készíteni.
- ▶ A beágyazott nyelv kiértékelésének miként akár teljesen független is lehet (vagyis például mohó nyelvet írhatunk le így).


# Funkcionális előnyök: Lusta kiértékelés

A "Do not Repeat Yourself" (DRY) elv, modularitás hatékonyabb támogatása.

- ▶ A gyakori sémák könnyen kiemelhetők, újrahasznosíthatóak, alkalmasak a vezérlési szerkezetek leírására.
- ▶ A mohó nyelvekben a sémák alkalmazása nem minden esetben éri meg, mivel a kiértékelés miatt azok *nem olvadnak össze*.

$$\begin{aligned} \text{any} &:: (\alpha \rightarrow \text{Bool}) \rightarrow [\alpha] \rightarrow \text{Bool} \\ \text{any } p &= \text{or} \circ \text{map } p \end{aligned}$$
$$\begin{aligned} \text{or} &:: [\text{Bool}] \rightarrow \text{Bool} \\ \text{or} &= \text{foldr False } (\vee) \end{aligned}$$
$$\begin{aligned} \text{any} &:: (\alpha \rightarrow \text{Bool}) \rightarrow [\alpha] \rightarrow \text{Bool} \\ \text{any } p &= \text{foldr False } (\lambda x r . p\ x \vee r) \end{aligned}$$
$$\begin{aligned} \text{any} &:: (\alpha \rightarrow \text{Bool}) \rightarrow [\alpha] \rightarrow \text{Bool} \\ \text{any } p [] &= \text{False} \\ \text{any } p (y : ys) &= p\ y \vee \text{any } p\ ys \end{aligned}$$
$$\begin{aligned} \text{foldr} &:: (\alpha \rightarrow \beta \rightarrow \beta) \rightarrow \beta \rightarrow [\alpha] \rightarrow \beta \\ \text{foldr } f\ z [] &= z \\ \text{foldr } f\ z (x : xs) &= f\ x (\text{foldr } f\ z\ xs) \end{aligned}$$

# Funkcionális előnyök: Lusta kiértékelés (példa)



**type** *Door*

*initial* :: *Door*

*react* :: *Door* → *Event* → *Door*

*isLocked* :: *Door* → *Bool*

# Funkcionális előnyök: Lusta kiértékelés (példa)

```
data State = Opened | Closed | Locked deriving Eq
```

```
data Event = Open | Close | Lock | Unlock
```

```
type Door = State
```

```
initial :: Door
```

```
initial = Opened
```

```
react :: Door → Event → Door
```

```
react s e = case (s, e) of
```

```
  (Closed, Open) → Opened
```

```
  (Opened, Close) → Closed
```

```
  (Closed, Lock) → Locked
```

```
  (Locked, Unlock) → Closed
```

```
  _ → s
```

```
isLocked :: Door → Bool
```

```
isLocked Locked = True
```

```
isLocked _ = False
```



# Funkcionális előnyök: Lusta kiértékelés (példa)

**data** *Tree* = *Node State (Event → Door)*

**type** *Door* = *Tree*

*initial* :: *Door*

*initial* = *fetch trees Opened where*

*trees* = [*construct Opened*, *construct Closed*, *construct Locked*]

*construct s* = *Node s (fetch trees ∘ f s)*

*fetch t s* = *head [x | x@(Node s<sub>1</sub> \_) ← t, s<sub>1</sub> ≡ s]*

*f s e* = **case** (*s*, *e*) **of**

  (*Closed*, *Open*) → *Opened*

  (*Opened*, *Close*) → *Closed*

  (*Closed*, *Lock*) → *Locked*

  (*Locked*, *Unlock*) → *Closed*

  \_ → *s*

*react* :: *Door* → *Event* → *Door*

*react* (*Node \_ f*) *e* = *f e*

*isLocked* :: *Door* → *Bool*

*isLocked* (*Node Locked \_*) = *True*

*isLocked* \_ = *False*

# Beágyazás másként: Template Haskell

# Template Haskell: Bevezetés

```
{-# LANGUAGE TemplateHaskell #-}  
module Fibs where  
  
import Language.Haskell.TH  
  
fibs :: [Integer]  
fibs = 0 : 1 : zipWith (+) fibs (tail fibs)  
  
-- [| ... |]: Oxford bracket, quasi quote, „template” (Q monád)  
-- $x, $(...): splice, "evaluate at compile time"  
fibQ :: Int → Q Exp  
fibQ n = [| fibs !! n |]
```

```
GHCi> :set -XTemplateHaskell  
GHCi> :load Fibs.hs  
...  
GHCi> $(fibQ 22)  
17711
```

# Template Haskell: A típusozott *printf()* visszatér

```
{-# LANGUAGE TemplateHaskell #-}
module Printf (printf) where

import Language.Haskell.TH
import Language.Haskell.TH.Syntax

data Format = D | S | L String

-- parse "%d is %s"  $\rightarrow_{\beta}^*$  [D, L " is ", S]
parse :: String  $\rightarrow$  [Format]

gen :: [Format]  $\rightarrow$  Q Exp  $\rightarrow$  Q Exp
gen [] x      = x
gen (D : xs) x = [|  $\lambda n \rightarrow$  $(gen xs [| $x ++ show n |]) |]
gen (S : xs) x = [|  $\lambda s \rightarrow$  $(gen xs [| $x ++ s |]) |]
gen (L s : xs) x = gen xs [| $x ++ $(lift s) |]

printf :: String  $\rightarrow$  Q Exp
printf s = gen (parse s) [| "" |]
```

```
GHCi> :load Printf.hs
```

```
GHCi> $(printf "Error: %s on line %d." "unexpected" 42
```

```
"Error: unexpected on line 42.")
```

# Template Haskell: Mit lehet még csinálni?

- ▶ *Feltételes fordítás*: A különféle előfeldolgozási megoldások helyett egy megbízhatóbb, típusozott alternatíva.
- ▶ *Önreflexió*: A programnak lehetősége van a saját belső szerkezetének elérésére, elemzésére.
- ▶ *Procedurális programszerkesztés*: Olyan metaprogramok írásának támogatása, amelyek azt adják meg, miként kell egy nála egyszerűbb programot előállítani, pl. *printf*.
- ▶ *Továbbfejlesztett absztrakciók*: Magasabb rendű függvények leképezése elsőrendű függvényekre, vagy egész számokkal indexelt függvények (*zip1*, *zip2*, ..., *zipN*) előállítása típusosan.
- ▶ *Optimalizációk, kiterjesztések*: Kísérleti kiterjesztések, problémafüggő optimalizációk megvalósítása a fordítóprogram módosítása nélkül.

# Template Haskell: Függvényépítő függvények (példa)

```
sel :: Int → Int → Q Exp
sel i n = [ λ x . $(caseE [ x ] [alt]) ] where
  alt :: Q Match
  alt = match pat (normalB rhs) []
```

```
pat :: Q Pat
pat = tupP [ varP a | a ← as ]
```

```
rhs :: Q Exp
rhs = varE (as !! (i - 1))
```

```
as :: [Name]
as = [ mkName ("a" ++ show i) | i ← [1..n] ]
```

$\$(sel\ 2\ 3) \longrightarrow$	$\$(sel\ 3\ 4) \longrightarrow$
$\lambda x . \mathbf{case}\ x\ \mathbf{of}$	$\lambda x . \mathbf{case}\ x\ \mathbf{of}$
$(a_1, a_2, a_3) \rightarrow a_2$	$(a_1, a_2, a_3, a_4) \rightarrow a_3$

```
GHCi> let sel_3_4 = $(sel 3 4)
GHCi> :type sel_3_4
sel_3_4 :: (t, t1, t2, t3) -> t2
GHCi> sel_3_4 ('a', 'b', 'c', 'd')
'c'
```

# Template Haskell: Több függvény gyártása (példa)

```
genSels :: Int → Q [Dec]
genSels n = declare [ s j i | i ← [2..n], j ← [1..i] ] where
  declare [] = return []
  declare (d : ds) = do
    x ← d
    xs ← declare ds
    return (x ++ xs)

s x y = do
  let name = mkName $ "sel" ++ show x ++ "." ++ show y
  body ← sel x y
  return [FunD name [Clause [] (NormalB body) []]]
```

*Sels.hs* (külön modul):

```
import Language.Haskell.TH
import Sel

$(genSels 10)
```

# A Quotation (Q) monádról

```
GHCi> import Language.Haskell.TH.Syntax
```

```
GHCi> import Language.Haskell.TH.Lib
```

```
GHCi> :type runQ
```

```
runQ :: Quasi m => Q a -> m a
```

```
GHCi> runQ [| 1 + 2 |]
```

```
InfixE
```

```
  (Just (LitE (IntegerL 1)))
```

```
  (VarE GHC.Num.+)
```

```
  (Just (LitE (IntegerL 2)))
```

```
GHCi> :{
```

```
  $(return (InfixE
```

```
    (Just (LitE (IntegerL 1)))
```

```
    (VarE (mkName "+"))
```

```
    (Just (LitE (IntegerL 2)))
```

```
  ))
```

```
:}
```

```
3
```

```
GHCi> :type [| 1 + 2|]
```

```
[| 1 + 2 |] :: Q Exp
```

```
GHCi> :type [d| x = 5 |]
```


```
[d| x = 5 |] :: Q [Dec]
```

```
GHCi> :type [t| Int |]
```

```
[t| Int |] :: Q Type
```

```
GHCi> :type [p| (x,y) |]
```

```
[p| (x,y) |] :: Q Pat
```





# Quasi Quotes

```
{-# LANGUAGE QuasiQuotes #-}
```

```
data E = EInt Int | EAdd E E | EMetaVar String | ...
```

```
pExp = spaced pTerm `chainl1` spaced op where
```

```
  pTerm    = pNum <|> pMetaVar
```

```
  pNum      = do { x ← digit; return (EInt $ digitToInt x) }
```

```
  pMetaVar  = do { char '$'; v ← ident; return (EMetaVar v) }
```

```
  ident     = do { c ← small; cs ← many idchar; return (c : cs) }
```

```
  op        = do { char '+'; return EAdd }
```

```
  ...
```

```
expr = QuasiQuoter
```

```
  { quoteExp = quoteEExpr, quotePat = quoteEPat
```

```
  , quoteDec = undefined, quoteType = undefined } where
```

```
  quoteEExpr s = do
```

```
    pos ← getPosition
```

```
    exp ← parseE pos s
```

```
    dataToExpQ (const Nothing) exp
```

```
  ...
```

Külön modul:

```
simplify :: E → E
```

```
simplify [expr | 0 + $x |] = x
```

```
main = print $ simplify [expr | 0 + 2 |]
```

# Template Haskell: Hibakeresés és önreflexió

```
GHCi> runQ [| 1 + 2 |] >=> putStrLn . pprint
1 GHC.Num.+ 2
GHCi> :set -ddump-splices
GHCi> $([| 1 + 2 |])
<interactive>:11:3-13: Splicing expression
    [| 1 + 2 |] =====> (1 + 2)
3
GHCi> :type id
id :: a -> a
GHCi> $(stringE . show =<< reify 'id)
...
"VarI GHC.Base.id
  (ForallT [PlainTV a_1627399870] []
    (AppT (AppT ArrowT (VarT a_1627399870)) (VarT a_1627399870)))
    Nothing
    (Fixity 9 InfixL)"
GHCi> $(stringE . show =<< reify ''Bool)
...
"TyConI (DataD [] GHC.Types.Bool []
  [NormalC GHC.Types.False [],NormalC GHC.Types.True []] [])"
```

